
一、TCL 簡介
TCL的全名為Tool Command Language，唸作”Tickle”，事實上它是一個Scripting Language（俗稱劇本語言或腳本語言），也是一個直譯器(Interpreter)。TCL 語言有三個特色：
1. 語法簡單，容易上手
2. TCL 的身份如同UNIX裡的Shell languages像是Bourne Shell (sh)、C Shell (csh)、Korn Shell (ksh) 與 Perl一樣，用來執行與控制系統上的程式。TCL具備足夠的程式化能力 (variable、flow control、procedure) 與存取檔案、程序 (Process) 及網路的功能，供組裝既有軟體元件以建立符合需求的新工具。
3. 可內嵌 (embed) 到應用程式中，讓軟體使用者透過程式員提供的高階 TCL 指令，自訂應用程式的行為。

除了上列三個主要特色外，底下所列的幾點也是 TCL 語言成功的原因：
· 跨平台，可在各種系統 (UNIX、Windows、Macintosh 等) 執行 TCL 程式
· 強大的字串處理能力『常規表示式 (Regular Expressions)』，協助程式員使用表示式的規則或樣式 (pattern)，用來搜尋、比對、粹取或是取代符合樣式的複雜字串。

· 可用 C 語言開發 TCL Extension Library，擴充 TCL 的指令與能力，例如著名的 Tk extension 讓 TCL 擁有 GUI Programming 的功能、expect extension用來開發可與應用程式互動式溝通的程式。

· 已有一大堆可在各式場合發揮效用的 package，諸如各種網路應用(ftp, http、email, dns, msn, icq 等)、圖形化介面開發 (TK, BWidget, Tktable, SpecTCL, VisualTCL, ActiveState KOMODO)、物件導向程式開發 (incrTcl, XOTcl, SNIT 等) 以及資料庫程式開發 (MetaKit, daFT 等)。
總結來說，TCL 語言就像是一瓶膠水一般，只要收集可資運用的應用程式與 TCL Package，在這些介面上塗塗抹抹，寫個簡單的 TCL Script (TCL 指令的集合)，在短暫的時間內就可組裝出一個新工具來。因此寫 TCL Script 開發軟體的過程就像是在玩樂高積木一般。
二、TCL 語法
TCL Script 可視為一個包含許多 TCL 指令 (TCL command) 的程式，一個 TCL 指令的基本語法為：
command arg1 arg2 arg3 ….
· 在 TCL 語言中，每行指令的第一個單字為指令名稱
· 空白字元用來分隔指令名稱與各個個別的參數
· 個別指令的分隔是以分號與換行符號作為分隔符號
· 一個參數如果超過一個英文單字 (亦即字串中間穿插空白或其它符號)，可用雙引號與大括號將這些元素組成 (grouping) 單一一個參數。使用雙引號與大括號的差別，在於TCL針對雙引號中的字串會做置換處理 (substitution)，例如變數值的代換或是執行包含於字串中的 TCL 指令(使用中括號來表示)，而對於大括號所括住的內容 TCL則不會有這些置換處理。
底下是一個簡單的 TCL Script 與執行後的輸出：
# Demo1.tcl

puts stdout one; puts stdout two

set x 4

set y 6

puts "$x + $y = [expr $x + $y]"

puts {$x + $y = [expr $x + $y]}

puts "Hello\n\nTCL!"
% ./Demo1.tcl

one

two

4 + 6 = 10

$x + $y = [expr $x + $y]

Hello

TCL!
下表所列的符號在 TCL 語言裏有特殊的意義：
	置換符號 (substitution symbols)

	$
	變數值置換符號。$ 符號用來取出指定變數的值。底下利用 set 指令設定 x 變數的值，並用 puts 指令輸出 x 的內容到 Console：
set x 5

puts stdout $x

如上列最後一行程式碼，取出變數值時需在變數名稱前加上 $ 符號。

	[ ]
	命令置換符號。TCL 將中括號中的內容視為一個指令，會執行其中的指令並將結果傳回。例如 Demo1.tcl L5：
puts "$x + $y = [expr $x + $y]"
在做完 x 與 y 兩個變數的置換後，TCL解譯器遇到中括號認為其中為另一個可執行的指令，指令名稱為 expr。在執行 expr指令前，TCL 先將 x 與 y 值置換後，傳遞給 expr 指令並呼叫執行之。expr 指令用來做數學式子的運算，會負責剖析參數並做數學運算。

	群組 (grouping symbols)

	“”
	雙引號可將多個元素組成單一一個參數，引號內的內容會被 TCL 進行置換處理，包括變數置換與命令置換。

	{ }
	大括號的功能與雙引號相同，但 TCL 不會對括號中的內容做任何解釋或處理，會照將括號中的內容視為一個參數，照本宣科原原本本地傳遞給指令去處理。例如 Demo1.tcl L6：

puts {$x + $y = [expr $x + $y]}
因為 TCL 對大括號中的內容不做任何處理，因此它的輸出為 $x + $y = [expr $x + $y]。另外，TCL 不會將大括號中的換行符號視為一個指令的結束，如果一個指令的參數很長時，我們就可以利用大括號這樣寫，把換行符號當成是參數的一部份：
puts {$x + $y =

  [expr $x + $y ] 

}

	其它

	\
	和 C base 語言一樣，反斜線可將特殊的字元做跳脫處理。例如 \n 代表換行符號。在 TCL 語言裏，如果一個指令超過一行，亦可在行尾使用反斜線做為續行的符號 (VB 指令如果超過一行，則是在行尾用底線符號代表續行)。

	分號與換行號
	代表一個 TCL 指令的結束符號。

	#
	註解符號。通常我們會在一行程式碼的第一個字元打上 # 符號，以代表本行程式為註解。注意！如果在 # 符號前有 TCL 指令，必須像這樣做：
puts hello ;# here is comment

亦即，在 # 符號前加上一個分號代表前面指令的結束。如果不這麼做，那TCL 便會把 # 視為一個參數丟給 puts指令去做處理。


變數
TCL 變數的可以隨時建立不用事先宣告，變數名稱的取法也沒有任何限制，所以可以使用任何字元來為變數命名，甚至變數名跟指令名稱相同也沒問題。使用時只有一點要注意的，就是 TCL 是大小寫有別的語言。通常，我們以 set 指令來設定變數值：
set x 5

set y 6

set PI 3.14
set PI*2 6.28   ;# 注意！變數名稱是 PI*2
puts $PI*2
set PI
TCL 的 set 指令除了設定變數值的用途外，也可以用來取出變數的內容，如上列程式最後一行使用 set 指令取出 PI的值 (注意！這邊不用在變數名稱前加上 $ 符號)。
unset 指令用來刪除一個或多個變數，來釋放記憶體空間：
unset x y PI

我們可以使用 info exists 指令來檢查變數是否存在。例如 incr 指令使用時，變數必須先建立才能為變數進行加法運算。此時便可使用 info exists 先檢驗變數的存在：
if {![info exists counter]} {

    set counter 0

} else {

    incr counter     ;# counter 加 1
}
TCL 在執行時會建立一些全域變數，可以利用 info global 列出有哪些全域變數：
% info global
argv argv0 tcl_version tcl_interactive var auto_oldpath errorCode auto_path errorInfo unknown_handlers unknown_handler_order auto_index env tcl_patchLevel argc tcl_libPath _ tcl_platform tcl_library
其中幾個比較常用的，像是 argc 存放程式的命令列參數個數，argv0 是目前執行的 TCL Script 名稱，argv 則是存放所有命令列參數的list；env 陣列存放系統的環境變數，而tcl_version 會告訴你目前的 TCL 版本為何。
置換與群組處理 (Substitution and Grouping)

底下是 TCL 置換處理與群組的幾點規則整理：
· TCL 指令的參數是以空白字元分隔，可以使用雙引號或大括號將多個元素組成一個參數。
· 如果是以大括號組出一個參數，TCL 不會對括住的內容進行任何置換，在對應的結束大括號出現前，任何字元都將視為參數的一部份，包含換行符號、分號或是內部巢狀的大括號。
· 如果是以雙引號組出一個參數，則在對應的結束雙引號出現前，雙引號中的內容會被 TCL 進行置換動作。

· 如果參數或是雙引號內的單字是以變數符號 $ 為開頭，則 TCL 為會進行變數的置換。

· 如果參數或是雙引號中有中括號，則在對應的中括號出現前，TCL 會對中括號中的內容進行指令的置換。
· 因為 TCL 是以空白字元作為參數分隔符號，因此要避免下列的錯誤：
if {$x > 1}{puts $x}
        ^ 在此加上一個空白字元，隔開 if 指令的第二及第三個參數。
TCL 的數學運算子
	- ~ !  
	「減號(Unary minus)」 / 「NOT位元運算(Bit-wise not)」 / 「NOT布林邏輯運算(Logical not)」。這些運算不可以用來操作字串(string)運算元，而且「NOT位元運算」只限於整數的操作。

	* / %  
	「乘(Multiply)」 / 「除(divide)」 / 「餘數(remainder)」。 這些運算不可以用來操作字串(string)運算元，而且「餘數」運算只限於整數的操作。

	+ -    
	「加(Add)」與「減(subtract)」。限用於數值運算元。

	<< >> 
	「左右移位運算 (Shift Left / Right)」。運算只限於整數的操作。

	< > <= >= 


	布林運算「小於(less)」/「大於(greater)」/「小於等於(less than or equal)」/「大於等於(greater than or equal)」。如果條件成立這些運算子會產生「1」的結果，否則產生「0」。 

	== != 


	布林運算「等於(equal)」/「不等於(not equal)」。每個運算會產生0/1的結果。可適用於任何運算元。

	&     
	「AND位元運算(Bit-wise and)」。限於整數的操作。

	^     
	「XOR位元運算(Bit-wise exclusive or)。限於整數的操作。

	|     
	「OR位元運算(Bit-wise or)。限於整數的操作。

	&&    
	「AND布林邏輯運算(Logical and)」。如果參與的兩個運算元都非零的話會得到1的結果，否則的話會得到0。 限用於數值運算元（不限整數或小數）。

	||    
	「OR布林邏輯運算(Logical or)」。如果參與的兩個運算元都是零的話會得到0的結果，否則的話會得到1。 限用於數值運算元（不限整數或小數）。

	x ? y : z 
	如果 x 為真時傳回 y 否則傳回 z


三、資料型態

TCL 語言的基本資料型態有：
· string

· list

· array

· handle：用於 I/O channel、socket, thread 等，此節暫不說明。
String 字串資料型態

字串是 TCL 語言最基本的資料型態，常見的字串處理指令有：string、append、format、scan 以及 binary。例如使用string 指令來計算指定字串的長度：
% set name “Brent Welch”
% string length $name

=>11

其中 string 指令的第一個參數代表對字串的操作方式。可以試著傳遞一個錯誤的參數給 string 指令，透過這個方式查看string指令有哪些可用的方法：
% string rick

bad option "rick": must be bytelength, compare, equal, first, index, is, last, length, map, match, range, repeat, replace, tolower, toupper, totitle, trim, trimleft, trimright, wordend, or wordstart
下表總結 string 指令的用法：

	string bytelength str
	傳回字串的位元組數(以UTF-8 encoding計算)，傳回值可能會跟計算字元數的string length有所不同

	string compare ? -nocase? ?-length len?str1 str2
	比較兩字串的內容，若相同傳回『0』、其他的傳回『1』。

-nocase：表示不分大小寫

例：

% string compare –nocase Rick rick

=>0

-length：可以指定要比較的字串長度

例：string compare –length 3 rick ricp

=>0

	string equal ?-nocase? str1 str2
	比較str1和str2的內容，若相同傳回『1』、否則傳回『0』

	string first subString string
 startIndex
	傳回 string 中相符子字串第一個出現的位置，若無則傳回『-1』。startIndex 代表從字串指定位置開始搜尋子字串。
例：

% string first rick ilovericktoo

=>5 

	string index string charIndex
	傳回index位置所在的字元，index的計算從『0』開始，使用『end』代表最後一個字元或使用『end-N』代表相對於最後一個字元的位置。

例：

% string index rick 2

=>c
% string index rick end

=>k
% string index rick end-2

=>i

	string last string
	傳回 string 中最後一個與子字串相符的位置，若無則傳回『-1』。startIndex 代表從字串指定位置開始搜尋子字串。
例：

% string frist rick rickANDrick
=>0
% string last rick rickANDrick

=>7

	string length string
	傳回字串的字元數

	string map ?-nocase? charMap string
	依據 charMap 中的字串對應表，更換字串中的內容。
例：

% string map {i c} rick

=>rcck
% string map {ri ja} rick

=>jack

% string map {r j i a} rick
=>jack

	string match ?-nocase? pattern str
	將字串拿來與pattern比對，若是相同則傳回『1』，若不同則傳回『0』。字串比對的方式為 glob style pattern match：
?：表示一個任意字元

*：表示隨意數量的任意字元。

[ ]：表示中括號中集合的一個任意字元，[abc]定義符合 abc 任意一個字元，[a-z] 定義符合所有小寫英文字母任意一個字元。
\?：表示一個問號字元。
例：

% string match tcl* tcltk

=>1

% string match tcl* rick

=>0
『?』表示一個任意字元。

例：

% string match ric? rick

=>1

% string match ri?? rick

=>1

% string match ric? rickpeng

=>0
『[ ]』表示在中括號集合中任意一個字元。

例：

% string match {[a-z]} rick

=>0

% string match {[a-z][a-z][a-z][a-z]} rick

=>1

	string range str i j
	取出str字串中從i到j位置的字元。可用 end 字元代表最後一個字元。

例：

% set x [string range iloverick 2 5]

=>over

	string repeat str count
	將str字串重覆印出count所表示的次數。

例：

% string repeat rick 3

=>rickrickrick

	string replace str first last ?newstr?
	以空字串或 newstr取代指定範圍的字串。First與last分別為所要取代的字串其頭、尾位置，newstr則為所欲代換成的字串，newstr 預設為空字串代表要截斷指定範圍的字串。

例：

% string replace iloverick 1 4 hate

ihaterick

	string tolower string ?first? ?last?
	將所定義的字串範圍中之字元轉換為小寫字元。

例：

% string tolower ILOVERICK 1 4

=>IloveRICK

	string totitle string ?first? ?last?
	將所定義的字串範圍中第一個字母轉換成大寫。

例：

% string totitle iloverick 1 4

=>iLoverick

	string toupper string ?first? ?last?
	將所定義的字串範圍中之字元轉換為大寫字元。

例：

% string toupper iloverick 1 4

iLOVErick

	string trim string ?chars?
	去除指定字串頭、尾中包含在chars中定義的任何字元。chars系統預設為空白字元。
例：

% string trim ililoverick rick

lilove
% string trim kkkloverickkkkk rick

=>love

	string trimleft string ?chars?
	去除在string左側包含在 chars中定義的任何字元，chars系統預設為空白字元。

例：

% string trimleft iloverick i

=>loverick
% string trimleft kkkloverickkkk lk

=>overickkkk

	string trimright string ?chars?
	去除在string右側包含在 chars中定義的任何字元，chars 系統預設為空白字元。

例：

% string trimright iloverick rick
=>ilove

	string wordend str index
	傳回指定 str 字串中指定位置上的單字的下一個位置。

例：

% string wordend {I love rick} 5
=>6

	string wordstart str index
	傳回str字串中指定位置上單字的起始位置。

例：

% string wordstart iloverick 9

=>0


string 指令注意事項
· 字串的比較儘量使用 string compare 或 string equal，例如：
if {[string compare $s1 $s2] == 0} {

    # s1 and s2 are equal

}

if {[string equal $s1 $s2]} {

    # s1 and s2 are equal

}

避免使用 == 來比較字串，雖然下列程式碼中，比較的兩個字串有不一樣的內容，仍會輸出 ack 的訊息，這是因為 TCL 會試圖將字串轉換為數字再進行比較，於是16進位的0xa會等於十進位的10，這可不是我們所期望的結果，因此建議儘量使用 string compare 或 string equal 來取代這種寫法：
if { “0xa” == “10”} { puts “ack” }
append 指令
append 指令用來將新的項目附加到指定變數的內容後。
例：

% set foo rick

rick

% append foo i love you

rickiloveyou
format 指令
與C語言中的printf十分相似的指令，format指令依據指定的格式將字串格式化。

    format spec value1 value2 …
其中spec的參數包含了文字或是任何字元。而一般定義的關鍵字包含了六大部份：

· 位置指示(position specifier)

· 旗標(flags)

· 欄寬(field width)

· 精確度(precision)

· 字元長度(word length)

· 轉換字元(conversion character)

由於在格式定義的時候常會有空白夾雜其中，切記要使用雙引號或大括號將定義內容群組(Grouping)起來。

	d
	帶正負號整數(Signed integer)

	u
	無正負號整數(Unsigned interger)

	i
	帶正負號整數。表示成為hex(0x)或octal(0)

	o
	無正負號的八進位數值。(Unsigned octal) 

	x or X
	無正負號的十六進位數值。(Unsigned hexadecimal)，x表示為輸出小寫的結果。

	c
	把數字對映成為ASCII 字元

	s
	字串

	f
	浮點數，格式為a.b

	e or E
	浮點數，格式為科學符號，a.bE+-c

	g or G
	浮點數，格式為%f或%e，依實際長度取短的表示


位置指示的表示方法為i$，意思是直接取得第i個參數的值。參數的計數是從1開始。底下的範例透過位置指示直接取用 format 第二個參數的內容：

  % format {%2$s} one two three
=>two
因為在 TCL 中，$ 符號有特殊意義，上列指令以大括號抑制了格式化字串中 $ 符號的變數置換作用，所以 i$ 的功能得以正常演出。但如果格式化字串是以雙引號來做群組 (grouping)，我們必須利用反斜線仰制 $ 符號的變數置換：
  % format “%2\$s” one two there
=>two

格式化旗標

	-
	靠左對齊

	+
	顯示數值的正、負符號

	space
	

	0
	以0 補滿

	#
	遇到octal將字首填入『0』，遇到Hex時字首填入『0x』


scan 指令

與C語言中的scanf十分相似的指令，scan指令依據指定的格式化條件剖析字串並將結果放入變數中。

    scan string format ?varName? ?varName?  …
例：
    % scan “a 123 Rick” char num str
    % puts “char = $char num = $num str = $str”

    => char = 97 num = 123 str =rick

scan 指令與 format 指令非常相像，%c 會將某個字元的 ASCII 數值讀入變數中。
Binary 指令
    binary format template value ?value ….?

    Binary scan value template variable ?variable …?

這邊我們看看幾個 binary 指令常用的用法，至於詳細 template 的設定方法請參閱 manual page。
使用 c 為模板(template)，將數值 97 格式化為 ASCII字元：

% binary format c 97

=>a

使用 c 為模板(template)，將字元 6 的 ASCII 值讀進 var1 變數中：
% set input 6
% binary scan $input c var1
% set var1
=> 54
一次scan 多個字元碼到 list 中：
    % binary scan abcde “c3” list

    % set list
    => 97 98 99

    % lindex list 1    ;# 取出 list 中第 1 個元素
    => 98

換個方式，scan 到多個個別變數中：
    % binary scan abcde “ccc” x y z
    % puts “x = $x y = $y z = $z”
    % x = 97 y = 98 z = 99
List 串列資料型態

TCL 的 list 指令有 list, lindex, llength, lrange, lappend, linsert, lreplace, 

lsearch, lset, lsort, concat, join, and split 等，提供您將資料放入 list、取出 list 中的元素、計算 list 的元素數量、取代 list 中的指定元素等等功能。通常我們會搭配 foreach 的使用，將 list 中的元素一一取出做運算。 
然而，在表現比較複雜的資料結構時，list 並不是最好的選擇方案，在這些場合裏使用 TCL 的 array 型態會比 list 更加適合。
	list arg1 arg2…..
	將 list 指令的參數將建構成一個 list。
例：

% list rick test

=>rick test

用 set 指令建構一個list：
% set mylist {rick test}
=>rick test

	lindex list i
	傳回在list之中第i個項目的內容。可以指定多個 index，以取出巢狀 list 中的項目。
例：

% set x {r i c k}

=>r i c k
% lindex $x 2

=>c

	llength list
	傳回在list之中的項目數。
例：

% set x {r i c k}

=>r i c k

% llength $x

=>4

	lrange list i j
	傳回list之中從i至j的項目。
例：

% set list {t p t s 1}

=>t p t s 1
% set y [lrange $list 1 2]

=>p t

	lappend listVar arg arg…..
	附加項目到listVar之後

例：

% set list rick

=>rick

% lappend list iloveyou

=>rick iloveyou

	Linsert list index arg arg..
	將資料插入在第i個項目之前。
例：

% set list {i love rick}

=>i love rick

% linsert $list 1 really

=>i really love rick

	lreplace list i j arg arg…
	將 list 中從i 到 j 個的項目取代為指定的參數。lreplace　傳回取代後的新 list。
例：

% set list {i love rick}

=>i love rick

% set list2 [lreplace $list 1 1 hate]
=>i hate rick

	lsearch ?options? list pattern
	傳回與 pattern 比對相符合的第一個元素位置，傳回 -1 如果沒有相符的元素。預設的比對方式為 Glob。
例：

% set list {i love rick}
=>i love rick

% lsearch $list love

=>1

	lset listVar index 
?index…? value
	使用 value 設定 list 第i 個元素的值。
例：

% set list {i love rick}

=>i love rick

% lset list 1 hate

=>i hate rick 

	lsort ?switches? list
	根據switch 所指定的選項為 list 排序。可用的選項有-ascii, -dictionary, -integer, -real, -increasing, -decreasing, -index ix, -unique, -command command.。
例：
% set list [list maa cxlin ognoc]

% lsort $list
=>cxlin maa ognoc

	concat list list …
	將多個 list 串成一個。
例：
% set list1 [list 1 maa]
% set list2 [list 2 cxlin]

% set list3 [list 3 ognoc]

% set listAll [concat list1 list2 list3]

% llength $listAll

=>6

	join list ?joinString?
	將list中的項目合併成一個字串。可以指定 joinString 作為每個項目間的分隔，預設的分隔字元為空白。
例：

% set list [list i love rick]
% join $list
=>i love rick

% join $list ###
=>i###love###rick

	split string 
?splitchars?
	將字串分割成一個list。可以指定 splitChars 作為字串中每個項目的分隔識別符號。
例：

% set list [split ”a b c”]
% length $list

=>3

% set list [split “a#b#c” “#”]

=>a b c

% lrange $list 0 end

=>a b c


通常我們會把 list 與 foreach 結合運用。例如底下的 foreach迴圈列出環境變數的名稱及內容(array names 可取出指定陣列的所有 index)：
    % foreach index [array names env] { 
    %     puts “$index = $env(index)”
    % }

    =>OS = Windows NT
    =>windir = C:\Windows

    =>ComSpec = C:\WINDOWS\system32\cmd.exe

    =>(略)

利用 foreach 與 list 寫個 join 指令：
proc join {list sep{} } {

   set s {}  ;# s is the current separator

   set result {}

   foreach x $list {

      append result $s $x

      set s $sep

   }

   return $result

}
Array 陣列資料型態

TCL 的array 和 Perl 的 associative array 類似，陣列是以字串作為索引。陣列在 TCL 語言中舉足輕重，許多資料結構都是以陣列為基礎設計而成。陣列元素內容的設定一樣是使用 set 指令：
% set price(apple) 10

% set price(orange) 12

% set quantity(apple) 5

% set discount(apple) 0.8

puts $price(apple)
=>10
	array exists arr
	判斷 arr 是否為一個陣列。傳回 1 代表是陣列。
例：

% set price(apple) 10
% array exists price

=>1

	array get arr 

?pattern?
	將 arr 的索引及元素值交錯建立成一個 list 後傳回。可以使用樣式(Pattern)來做比對。
例：

% set price(apple) 10
% set price(orange) 12

% array get price
=>orange 12 apple 10

	array names arr 

?mode? ?pattern?
	傳回 arr 的索引list。可以使用 mode 指定樣式的比對方式 (可為 –exact、-glob(default) 或 –regexp)。
例：

% array names price
=>orange apple

	array set arr list
	根據list的內容建構一個陣列。
例：

% array set price [list apple 10 orange 12]
% array get price

=>orange 12 apple 10

	array size arr
	傳回 arr 的大小。
例：

% array size price
=>2

	array unset arr 

?pattern?
	釋放符合樣式的陣列元素。如果沒有指定樣式，則會釋放整個陣列。
例：
% array set price [list apple 10 orange 12]

% array get price

=>orange 12 apple 10

% array unset price app*
% array get price

=>orange 12

% set price(apple) 10
% array get price

=>orange 12 apple 10
% array unset price

% set price
=>can't read "price": no such variable

	array startsearch arr
	起始陣列的搜尋功能，並傳回一個標記(token id)。

	array nextelement arr id
	在陣列token-id所標記的搜尋中，將下一個元素的索引傳回。沒有元素時傳回空字串。

	array anymore arr id
	檢查陣列token-id所標記的搜尋中，是否還有元素可取用，還有元素傳回1否則傳回0。

	array donesearch arr id
	結束 id 所識別的搜尋。

	array statistics arr
	傳回arr陣列雜湊表(hash table)的統計表。


陣列也常拿來與 foreach 迴圈搭配使用，例如：
% set price(apple) 10

% set price(orange) 12

% array get price

=>orange 12 apple 10
% foreach {key value} [array get price] {

     “price($key) = $value”
}

=>price(orange) = 12

=>price(apple) = 10
或者這樣寫，程式的結果也會一樣：

% foreach key [array names price] {
    “price($key) = $price($key)”
}
因為使用 array get 或foreach 時，TCL 的作法是先產生一個暫時性的list，這樣會浪費一些記憶體空間與執行速度。改用 array 的搜尋功可以加快陣列元素巡訪(iterate)的速度：
% set searchToken [array startsearch price]
% while {[array anymore price $searchToken]} {

    set key [array nextelement price $searchToken]

    set value $price($key)

    puts “$key = $value”
 }

 array donesearch price $searchToken

=>orange = 12
=>apple = 10

四、控制結構
如同先前TCL語法一節所介紹的，對TCL來說每條敘述(Statement)都只是一個指令與傳遞給指令的參數，其中每個指令第一個英文單字為指令的名稱。本節所要介紹的控制結構，包括迴圈指令：while、for、foreach；條件判斷用的指令：if及switch；錯誤處理用的指令：catch；以及扮演配角一職用來調整控制結構用的指令：break、continue、return與error等等，事實上在TCL 語言中，這些控制結構也都是指令。
這些控制結構通常在某些條件成立時，會執行其指令主體的程式碼(command body)，例如：
set x 2
set y 3

if {$x < $y} {

    puts “x is less than y!”

}
我們知道在TCL語言中，大括號可以把多個元素(也就是很長的字串)群組起來變成一個參數。所以，以指令的觀點來看上列的 if 指令可知，if 指令接受兩個參數，其中第一個參數是一個條件敘述，只有在此條件成立的情況下，指令主體(command body)，也就是if指令的第三個參數，才會被if指令當成是一段TCL Script把它觸動執行之。
我們也知道換行符號與分號是代表一個指令的結束，如果一個指令的參數太長時，可以利用大括號括住參數，把換行符號也當作是參數的一部份。在控制結構中，指令主體的寫法通常都會和上列的if指令一樣，使用大括號括住。這樣做有兩個好處，第一是我們可以在指令主體中寫很多行敘述，而這些敘述會被大括號群組起來變成一個參數傳給if指令，第二則是使用大括號可以避免TCL解譯器在執行if指令前，就先進行置換的處理，產生非預期的結果。
If Then Else

If 指令是最基本的控制結構，它的語法是：
    if expression ?then? body1 ?else? ?body2?
if 指令在條件測試 (expression) 的結果為真是執行指令主體 (body1)，否則執行另一個指令主體 (body2)。例如：
if {$x > 0} {

    puts “x is greater than zero!”

} else {

    puts “x is less than zero!”

}

你可以省略then跟else兩個關鍵字，但像上面的寫法一樣我們通常只省略then。另外在條件測試和指令主體兩參數，建議養成習慣用大括號將它他們括住。當你需要判斷一連串的條件時，可以使用if搭配elseif 指令：
if {$temperature < 24} {
    puts “It’s cool!”

} elseif {$temperature < 30} {

    puts “It’s a little hot!”
} else {

    puts “It’s quite hot!”

}

然而，如果要判斷的條件式很多，可以選擇使用底下要介紹的switch指令。
Switch
switch 指令可用來比對許多樣式 (pattern matching)，並執行符合樣式字串的指令主體。switch 的語法為：
switch flags value pattern1 body1 pattern2 body2 …?default defaultBody?
只有第一個樣式比對符合的指令主體會被執行，其它則會被忽略。可以在switch指令最後加上一個default樣式，代表如果沒有任何一個樣式比對符合的話，就執行default的指令主體。

你可以設定switch指令的flags，指定字串的樣式比對方法：
-exact：使用完全相符的樣式比對方法。即要比對的 value與比對的樣式字串完全相同才算比對成功。
-glob：使用glob style 的樣式比對方法。請參考資料型態一節有關 string match的說明。glob style 樣式比對是switch 指令的預設比對方法。
-regexp：使用『常規表示式 (Regular Expression)』的樣式比對方法。有關常規表示式請見後述。
--：代表switch指令的flag設定已結束，接著的下一個字串是要被比對的字串 (value)。使用 -- 允許value是以 - 符號開頭的字串。通常我們會在value字串前加上這個flag。
另外，值得注意一點是，switch樣式比對有無使用大括號括住的影響，我們直接從底下的範例程式及執行結果來看：
set x 5

set y 5

switch -glob -- $x \
    $y { puts “x = y” } \
    {[0-9]} { puts “ x = 5” } \
    default { puts “x > 10” }

switch -glob -- $x {
    $y { puts “x = y” }
    [0-9] { puts “x = 5” }
    default { puts “x > 10” }

}

執行結果為：
x = y
x = 5

我們知道，使用大括號會抑制TCL的置換處理 (substitution)，第一個switch指令因為沒有使用大括號，因此TCL會進行可能的置換處理 (包括變數置換或命令置換)，因此 $y 會被置換成 5，這也表示我們可以使用變數來存放樣式供比對；另外因為中括號中的內容TCL會視為一個指令，因此我們使用大括號將[0-9]樣式括住以抑制此處的置換。由於switch指令在比對x的值時，發現與y變數置換後的值相同，因此輸出結果x = y。
第二個switch指令使用大括號，因此大括號中的內容不會有置換處理。所以switch指令中，第一個樣式 $y 代表著只有在x的值為「$y」時，才會比對成功。也由於沒有置換動作，因此我們不用擔心TCL會將中括號中的內容視為一個指令，所以可以使用中括號直接寫 [0-9] 做為樣式。由於x的值為5，符合 [0-9] 的樣式，所以輸出結果 x = 5。
如果你希望比較的字串，只要符合許多個樣式其中一個，就執行同樣的指令主體，那你可以這樣寫：
set var see
switch -glob -- $var {

    {see} -
    {saw} -
    {seen} { puts “All mean the same thing!”}

} 
=>All mean the same thing

While
指令語法：

while booleanExpr body

while 指令有兩個參數，一是條件測試，一是指令主體。while指令會不斷判斷條件敘述，並在條件測試的結果為真時，執行指令主體。下列的程式會開啟 autoexec.bat 檔，讀取檔案中的內容並輸出到Console中，程式中的gets指令會從指定的檔案讀取一行資料，並傳回讀取到的字元數，讀到檔案尾時傳回 -1：
set fileId [open c:/autoexec.bat r]

while {[gets $fileId line] >= 0} {

    puts stdout $line

}
close $fileId
在使用while指令時要注意TCL的置換處理，例如底下的程式會是一個無窮迴圈，因為 TCL在執行while指令前，已將 $i 代換為 0才傳遞給while指令作為參數，因此while指令的條件測試結果永遠為真：
set i 0 ; while $i<10 { incr I }
要避免請保持良好的寫碼習慣 (coding style)，使用大括號來避免TCL的置換，避免非預期的錯誤：
set i 0 ; while {$i < 10} { incr I }

For
指令語法：
for initial test final body

TCL 的for指令和C語言相類似，for 指令需要四個參數。第一個參數用來設定迴圈的起始狀態，例如計數值的起始設定，第二個參數用來做條件測試，第三個參數是迴圈指令主體 (body) 每執行一次後要做的事，通常我們會在這邊增減計數值。底下的範例使用迴圈印出三次目前的時間：

for {set x 1} {$x <= 3} {incr x } {

    puts [clock format [clock seconds] -format {%D %T} ]
}

=>06/22/04 22:04:10

=>06/22/04 22:04:11

=>06/22/04 22:04:12
其中clock seconds會以秒數為單位傳回目前的時間，clock format則用來將指定的時間進行格式化，本例的格式化字串為 {%D %T} 代表輸出的時間格式為 %m/%d/%y %H:%M:%S。另外after 指令會讓程式停留一段時間後繼續進行，停留時間的單位為千分之一秒 (millisecond)。
之前while指令讀取檔案內容的範例，可以用for指令改寫成 (可以看到此處在for指令的initial與final兩個參數都是空的)：
set fileId [open c:/autoexec.bat r]

for {} {[gets $fileId line] >=0} {} {

    puts stdout $line

}
close $fileId
使用for 指令來巡訪 (iterate)一個list，印出list中的內容：
for {set i 0} { $i < [llength $list] } {incr i} {
    puts [lindex $list $i]
}

但使用for指令巡訪list並不是好的寫法，它的效率不是很好，因為for指令需要不斷呼叫llength及lindex去取得list的狀態。比較好的方案，是選擇使用foreach指令。
Foreach
指令語法：
foreach listVar list body
foreach指令主要用於list的巡訪 (iterate)，來看一個數列加總的寫法：
set total 0
set numList [list 1 2 3 4 5 6 7 8 9 10]
foreach num $numList {
   set total [expr $total + $num]
}

puts stdout “total = $total”
=>total = 55
foreach指令也可針對一個list，設定多個迴圈變數來處理之，例如：
foreach {key value} {apple 10 orange 12} {

    puts “$key=$value”

}

=>apple=10

=>orange=12

對於陣列，可先讓array get指令將陣列的索引與值變成一個list後傳回，再拿給foreach指令運用：
set price(apple) 10

set price(orange) 12

foreach {key value} [array get price] { 

    puts “$key=$value”

}

=>orange=12

=>apple=10

break 與continue
break指令和continue指令用於迴圈結構中，break用來中斷迴圈，而continue則用來讓迴圈進行下一個重覆 (iteration)。
Catch
指令語法：
catch command ?resultVar?

當TCL指令發生錯誤時，例如使用open開啟檔案失敗，TCL會中斷目前執行的TCL Script。我們可以使用catch指令來補捉這些錯誤，catch指令會執行指定的指令，如果指令沒有發生錯誤，則catch指令會將指令的傳回值放入第二個參數，或是在指令發生錯誤時將錯誤訊息放入第二個參數。如果沒有捕捉到任何錯誤，catch指令本身的傳回值會是0，否則會傳回非0的數值。
我們將前面讀取檔案的內容加上catch指令，讓程式更嚴謹一些避免執行到一半便被TCL中斷執行：
if {[catch {open c:/autoexec.bat r} fileId ]} {
    puts “opening file error: $fildId”

} else {

    while {[gets $fileId line]} { 
        puts $line

    }

    close $fileId

}

五、Procedure
TCL 的程序跟一般程式語言的程序一樣，是程式模組化的工具，當你定義好一個procedure時，它就會變成一個TCL指令。TCL 使用proc指令來定義一個procedure：
proc procName argList body

proc指令第一個參數是procedure的名稱，第二個參數是procedure的參數列，第三個參數則是procedure的指令主體，會在procedure被呼叫時執行起來。關於程序的傳回值，除非你用return指令明示程序的傳回值，否則TCL會將程序最後一行指令的執行結果當做是程序的傳回值。
例如定義一個計算圓面積的procedure：

set PI [expr 2 * asin(1.0)]
proc c_area {rad} {

    global PI
    return [expr $PI * $rad $rad]
}

% c_area 3
=> 28.2743338823
其中我們也定義了一個全域變數PI，接受一個參數rad作為半徑。程式在procedure外使用expr及TCL內建的數學函式asin計算圓周率PI，並在c_area程序中使用這個全域變數來計算圓面積。值得注意的一點時，在程序的變數空間 (Scope) 預設是看不到全域變數的，如果要在程序中存取全域變數，則必須先使用global來宣告才行。
我們也可以指定程序參數的預設值 (default value)，例如底下修改後的c_area程序在沒有參數的情況下，會將rad參數預設值設成1：
set PI [expr 2 * asin(1.0)]

proc c_area { {rad 1} } {

    global PI

    return [expr $PI * $rad $rad]

}

% c_area 3
=> 28.2743338823

% c_area
=> 3.14159265359
我們也可以定義可接受不定參數的程序，透過args這個關鍵字 (TCL 會將接受到的參數變成一個list放入args變數中)：
proc printArgs { args } {
    puts “Total arguments: [llength $args]”

    puts “The argument list = [lrange $args 0 end]”

}

% printArgs

=>Total arguments: 0

=>The arguments list are:

% printArgs one two three
=>Total arguments: 3

=>The arguments list are: one two three

在TCL語言裏，我們甚至可以使用rename指令將一個指令重新命名，例如將TCL的unknown指令更名為originalUnknown：

rename unknown originalUnknown
接著我們可以自行定義unknown，替unknown訂定新的行為，有需要時再呼叫originalUnknown，這樣對原本呼叫unknown的使用者來說，會完全轉移到新定義unknown。另外，你也可以使用rename指令將既有的指令定義成空字串，如此一來，指令就會不存在。例如，我們將exec指令定義成空字串，這樣使用者便不能用exec執行其它程式：
rename exec {}

六、TCL 內建指令
下表條列 TCL 內建的指令及用途說明。
	指令
	說明

	after
	等後一段時間後執行指定的指令

	append
	附加參數的內容到指定變數之後

	array
	用來查詢陣列的狀態及搜尋陣列元素

	binary
	用來做字串與二進位資料的轉換

	break
	用來中斷迴圈

	catch
	補捉錯誤

	cd
	變更工作目錄

	clock
	用來取得時間及設定時間輸出的字串格式

	close
	用來關閉 I/O 串流(stream)

	concat
	用來將多個參數串成一個list，或是將多個list合成一個。

	console
	用來控制輸入命令用的主控台(Console)

	continue
	讓迴圈進行下一個重覆(iteration)

	error
	引發一個錯誤

	eof
	用來檢查檔案是否已到結尾

	eval
	將參數串接成一個字串後，將字串視為一個 TCL Script丟給TCL 解譯器去執行。

	exec
	用來執行一個程式

	expr
	用來做數學式子的運算

	fblocked
	用來檢查 I/O channel 資料是否已準備就緒供存取

	fconfigure
	用來設定 I/O channel 的屬性

	fcopy
	從一個 I/O channel 拷貝資料到另一個 I/O channel

	file
	用來查詢檔案系統的狀態

	fileevent
	用來註冊 I/O 事件發生時的 callback。

	flush
	用來清除 I/O 串流的 buffer

	for
	迴圈控制

	foreach
	迴圈控制

	format
	和 C 語言 printf 類似，用來為字串格式化。

	gets
	從 I/O 串流中讀進一行資料

	glob
	傳回符合樣式(pattern)的檔案名稱

	global
	全域變數在 procedure 中必須使用 global 來宣告它的存在，才能在procedure中存取全域變數

	history
	條列輸入過的指令記錄

	if
	控制結構。

	incr
	用來增加一個變數值

	info
	用來查詢TCL 解譯器的狀態

	interp
	用來產生另一個 TCL interpreter

	join
	用指定的分隔字串將list元素串成一個字串

	lappend
	附加元素到list之後

	lindex
	取得list中的元素

	linsert
	在list指定位置上加入一個元素

	list
	將list指令的參數建構成一個list

	llength
	傳回list中的元素數量

	load
	載入定義有TCL 指令的shared library

	lrange
	傳回list中指定範圍的元素

	lreplace
	取得list中指定範圍的元素

	lsearch
	傳回與指定樣式(pattern)比對相符合的第一個元素位置

	lsort
	為list排序

	namespace
	用於namespace的管理

	open
	用來開啟檔案

	package
	用來管理package

	pid
	傳回目前的process id

	proc
	用來定義TCL procedure

	puts
	輸出字串到一 I/O 串流中

	pwd
	傳回目前的工作目錄

	read
	從I/O 串流中讀入指定大小的資料

	regexp
	用來比對『常規表示式 (Regular Expressions)』

	regsub
	使用常規表示式來代換字串

	rename
	替TCL指令改名

	return
	用於procedure中的傳回值

	scan
	依據指定的格式化條件剖析字串並將結果放入變數中

	seek
	將I/O 串流的存取位置移動指定的偏移量

	set
	用來設定變數值

	socket
	用來建立一個TCP/IP連線

	source
	用來載入TCL Script檔案

	split
	用來將字串分割成一個list。

	string
	用來對字串進行各式運算

	subst
	對指定的字串進行置換處理(包括反斜線、指令及變數置換)

	switch
	控制結構

	tell
	傳回指定 I/O串流目前的存取位置

	time
	用來衡量指定TCL指令的執行時間

	trace
	用來追蹤變數的存取狀況

	unknown
	用來處理TCL解譯器不認識的指令

	unset
	用來刪除變數並釋放記憶體空間

	variable
	用來宣告namespace的變數

	vwait
	讓TCL進入事件迴圈中(event-loop)直到指定變數被寫入值為止

	while
	迴圈控制

